Eigenfunction Expansions

William O. Bray

The theme of this notebook is to illustrate the behavior of eigenfunction expansions. We consider the Sturm-Liouville problem

$$\text{ODE:}\ \ X'' + \mu X = 0,\ \ 0 < x < \pi$$
$$\text{BC:}\ \ X(0)=0,\ \ X'(\pi)+X(\pi)=0$$

The eigenvalues are all positive and writing $\mu = \lambda^2$, can be computed by solving the equation $\tan(\pi\lambda) = -\lambda$. Graphically, the solutions are illustrated in the following plot.

In[1]:=

```
Plot[{Tan[Pi x], -x}, {x, 0, 2 Pi},
 PlotStyle → {{Black, Thick}, {Green, Thick}},
 AxesStyle → Directive[12], PlotRange → {{0, 6}, {-8, 8}},
 Ticks → {{2, 4, 6}, {-5, 5}}]
```

Out[1]=



It is clear that $\lambda_n \approx n - \frac{1}{2}$. The first thirty values of λ are estimated as follows.

In[2]:=

```
lam = Table[
  x /. FindRoot[Tan[Pi * x] == -x, {x, (2 n - 1) / 2 + 0.005}], {n, 1, 30}]
```

Out[2]=

```
{0.787637, 1.67161, 2.61621, 3.58655, 4.56859, 5.55668,
 6.54824, 7.54196, 8.53712, 9.53327, 10.5301, 11.5275,
 12.5254, 13.5235, 14.5219, 15.5205, 16.5192, 17.5182,
 18.5172, 19.5163, 20.5155, 21.5148, 22.5141, 23.5135,
 24.513, 25.5125, 26.512, 27.5116, 28.5112, 29.5108}
```

The eigenfunctions are given by $X_n(x) = \sin(\lambda_n\, x)$. A plot of a few of these is given as follows.

In[3]:=
```
Plot[{Sin[lam[[1]] x], Sin[lam[[4]] x], Sin[lam[[7]] x]}, {x, 0, Pi}]
```

Out[3]=


NOTE: While each eigenfunction is a periodic function (on the real line), the collection of such does not have a common period as in the case of classical Fourier series.

Given a function $f \in L^2[0, \pi]$, the Fourier series of interest has the form:

$$f \sim \sum_{n=1}^{\infty} \hat{f}_n \sin(\lambda_n x), \qquad \hat{f}_n = \frac{1}{\Lambda_n} \int_0^{\pi} f(x) \sin(\lambda_n x)\, dx, \qquad \Lambda_n = \frac{2\pi\lambda_n - \sin(2\pi\lambda_n)}{4\lambda_n}$$

For later use the following commands are useful.

In[4]:=
```
nor = Table[
   (2 Pi lam[[n]] - Sin[2 Pi lam[[n]]]) / (4 lam[[n]]), {n, 1, 30}];
ps[N_, x_, coef_] := Sum[coef[[n]] Sin[lam[[n]] x], {n, 1, N}]
```

Example 1

Consider the piecewise continuous function

In[6]:=
```
g[x_] := If[x ≤ Pi / 2, 1, -1]
```

We can compute the Fourier coefficients the following way.

In[7]:=
```
gg[y_] := Evaluate[Integrate[g[x] Sin[y x], {x, 0, Pi}]]
ghat = Table[gg[lam[[n]]] / nor[[n]], {n, 1, 30}]
```

Out[8]=
```
{-0.29754, 1.14304, 0.415534, -0.0562663, -0.0642458,
 0.307578, 0.207724, -0.0313, -0.033097, 0.172391,
 0.135712, -0.0213567, -0.0221339, 0.119325, 0.100506,
 -0.0161674, -0.0166001, 0.0911549, 0.0797437, -0.0129971,
 -0.0132727, 0.07372, 0.066072, -0.010863, -0.0110538,
 0.0618741, 0.0563946, -0.00932952, -0.00946949, 0.0533039}
```

In[9]:=
```
Plot[{g[x], ps[8, x, ghat], ps[16, x, ghat], ps[24, x, ghat]},
 {x, 0, Pi}, Ticks → {{0, Pi / 2, Pi}, {-1, 0, 1}}]
```

Out[9]=



Example 2

In[10]:=
```
b = -Pi / (1 + Pi);
h[x_] := b x + Pi
```

In[12]:=
```
hh[y_] := Evaluate[Integrate[h[x] Sin[y x], {x, 0, Pi}]]
hhat = Table[hh[lam[[n]]] / nor[[n]], {n, 1, 30}]
```

Out[13]=
```
{2.12233, 1.10385, 0.734653, 0.545122, 0.431492, 0.356369,
 0.303226, 0.263733, 0.233266, 0.209067, 0.189392, 0.173086,
 0.159355, 0.147635, 0.137517, 0.128693, 0.12093, 0.114049,
 0.107908, 0.102393, 0.0974138, 0.0928956, 0.0887775, 0.0850086,
 0.0815463, 0.0783548, 0.0754034, 0.0726662, 0.0701205, 0.0677471}
```

In[14]:=
```
Plot[{h[x], ps[8, x, hhat],
  ps[16, x, hhat], ps[30, x, hhat]}, {x, 0, Pi}]
```

Out[14]=



Example 3

The following example illustrates the uniform convergence under the conditions that the function is C^2 and satisfies the boundary conditions.

In[15]:=
```
a = (1 - Pi - Pi^2) / (3 Pi^2 + Pi^3);
f[x_] := a x^3 + x^2 - x
```

In[17]:=
```
ff[y_] := Evaluate[Integrate[f[x] Sin[y x], {x, 0, Pi}]]
fhat = Table[ff[lam[[n]]] / nor[[n]], {n, 1, 10}]
```

Out[18]=
```
{0.32353, -0.440254, -0.0326619, -0.0375722, -0.00904663,
 -0.00925037, -0.00350829, -0.00351896, -0.00169098, -0.00168785}
```

In[19]:=

```
Plot[{f[x], ps[2, x, fhat], ps[4, x, fhat]}, {x, 0, Pi}]
```

Out[19]=